

A Quick Introduction to Handling Conflicts in Yacc Parsers

Saumya Debray
Dept. of Computer Science
The University of Arizona
Tucson, AZ 85721

September 8, 2005

1 How Yacc-generated Parsers Work

Yacc is a tool that takes a context-free grammar as input and produces (if it is able to) a parser for that grammar as output. A message from yacc indicating a “conflict” means that at some point in the course of generating a parser for its input grammar, it faced a choice between more than one possible action for the parser to take. While yacc uses default rules to choose one out of the multiple possible actions in such cases, the very presence of a conflict means that, potentially, something may be wrong and deserves careful examination.

Yacc produces parsers that belong to a family known as “LR parsers.” Intuitively, such parsers can be thought of as similar to finite state machines, in that they have states and transitions between states; one big difference, however, is that unlike the finite automata used in lexical analysis, the parser also has a stack that it uses for pushing and popping grammar symbols (it is this stack that allows the parser to go beyond regular languages and handle context-free languages). Broadly speaking, such parsers repeatedly perform two actions: *shift* and *reduce*. The general idea of the parsing algorithm is to match the input that has been seen with the right hand sides of grammar productions until an appropriate match is found, to replace it with the corresponding left hand side. To “match the input ... until an appropriate match is found,” the parser saves the grammar symbols it has found so far on its stack. A *shift* action takes the next token from the input and pushes it onto the stack. A *reduce* action happens when the sequence of grammar symbols on top of the parser’s stack matches the right hand side $x_1x_2 \cdots x_n$ of an appropriate grammar production

$$A \rightarrow x_1x_2 \cdots x_n$$

and the parser decides to “reduce” this to the left hand side of the production, i.e., pop the string $x_1x_2 \cdots x_n$ off the stack and push A instead.

The details of why and how these actions are chosen, and how they help with parsing, will be discussed in class, or may be found in a compiler text. *For our purposes, it suffices to understand the main intuition here: a shift action means that the parser is waiting to see more of the input before it decides that a match has been found; a reduce action means that a match has been found.*

There are two types of conflicts we might encounter: *shift-reduce* and *reduce-reduce*:

- *Shift-reduce conflict.* This occurs when the parser is faced with a choice of a shift action and a reduce action. (Yacc’s default action in the case of a shift-reduce conflict is to choose the *shift* action.)
- *Reduce-reduce conflict.* This occurs when the parser is faced with a choice of two different productions that could be used for a reduce action. (Yacc’s default action in the case of a reduce-reduce conflict is to reduce using the production that comes first, textually, in the input grammar specification.)

2 Parser States and the y.output File

In order to understand the source of a parser conflict, we need to know a little more about parser states. Intuitively, the parser processes a whole bunch of grammar rules “in parallel,” using the next input token to guide this processing at each stage. Each parser state, therefore, corresponds to one or more grammar rules in different stages of being processed. We’ll use a ‘•’ to mark the point upto which a given rule has been processed: e.g., given a rule ‘ $A \rightarrow \mathbf{a} \mathbf{b} \mathbf{c}$ ’ where the first two symbols in the right hand side have been processed, but not the third, we’ll write

$$A \rightarrow \mathbf{a} \mathbf{b} \bullet \mathbf{c}$$

Each parser state can be thought of as being labelled with a number of such “dotted rules” (usually called “items”). In order to figure out the reason for a conflict, we have to find out (1) which state has a conflict; and (2) the reason for the conflict.

The first step is to use the ‘-v’ option for `yacc`, which causes it to create a file `y.output` containing verbose information about the parser’s states. As a concrete example, consider the following simple grammar for generating strings that contain an even number of 0s:

Grammar 1:

$$\begin{aligned} S &\rightarrow 0 S 0 \\ S &\rightarrow \varepsilon \end{aligned}$$

The corresponding Yacc rules are:

```
S : '0' S '0'
  | /* epsilon */
  ;
```

On invoking `yacc`, we get the following:

```
% yacc -d -v grammar.y
yacc: 1 shift/reduce conflict
```

The `-v` option causes `yacc` to generate the file `y.output`, which contains, among other things, the following:

```
0 $accept : S $end
1 S : '0' S '0'
2   |
...
1: shift/reduce conflict (shift 1, reduce 2) on '0'
state 1
    S : '0' . S '0' (1)
    S : . (2)

    '0' shift 1
    S goto 3
...
```

The first three lines shown, which appear at the beginning of the `y.output` file, simply lists the grammar rules (rule 0 is an artificial one introduced by `yacc` for convenience; rules 1 and 2 are those appearing in the

input to yacc). After this, each parser state is shown together with the corresponding set of items (“dotted grammar rules”); here we show the one state that happens to have a conflict.

The following line tells us the location and cause of the conflict, with the relevant phrases shown in boxes:

```
1: shift/reduce conflict (shift 1, reduce 2) on '0'
```

The different parts of this line are to be understood as follows:

1. The ‘1’ at the beginning tells us which state has the conflict (state 1).
2. The next item specifies the conflicting actions:
 - (a) The ‘shift 1’ means that one action is “*shift, go to (i.e., stay in) state 1.*”
 - (b) The ‘reduce 2’ means that the other action is “*reduce using rule #2*” where the rule number is with respect to the list at the beginning of the `y.output` file, i.e., the rule ‘ $S \rightarrow \varepsilon$ ’.
3. The ‘on ‘0’ at the end indicates the input token on which this conflict arises: in this case, ‘0’.

This is followed by information about state 1. The first two lines after this tell us about the items in this state:

<u>Yacc output</u>	<u>“Dotted” notation</u>
S : '0' . S '0'	$S \rightarrow 0 \bullet S 0$
S : .	$S \rightarrow \bullet$

The final two lines refer to the parser’s actions in this state. The line ‘0’ shift 1’ means “*on input 0, shift and stay in state 1.*” The line ‘S goto 3’ refers to the state transition on a reduce action, and are not relevant for this discussion.

3 Diagnosing Conflicts

3.1 Shift/Reduce Conflicts

As shown above, the `y.output` file shows us the cause and location of the shift/reduce conflict. In the example above, we have:

```
1: shift/reduce conflict (shift 1, reduce 2) on '0'
```

Given our basic intuition about what items in a state as well as shift and reduce actions mean, this means that yacc is considering the following possibilities at this point If the next input token is ‘0’:

1. The ‘0’ in the input comes from the ‘S’ that this parser state is about to process (i.e., which is to the immediate right of the ‘•’), so the parser should shift this input token (and perhaps others) waiting to eventually reduce using the grammar rule ‘ $S \rightarrow 0 S 0$.’
2. The ‘0’ in the input is actually that at the end of the item ‘ $S \rightarrow 0 \bullet S 0$ ’, which means that the ‘S’ that the parser state is about to process should “disappear” via the production ‘ $S \rightarrow \varepsilon$ ’. In this case, the parser should carry out a reduce action using this rule.

3.2 Reduce/Reduce Conflicts

As the name suggests, reduce/reduce conflict arises when there are (at least) two different grammar rules that can be used for a reduce action in some parser state. As an example, consider the following variation on the grammar shown above:

Grammar 2:

$$\begin{aligned} S &\rightarrow T \\ S &\rightarrow \varepsilon \\ T &\rightarrow \mathbf{0} \mathbf{0} T \\ T &\rightarrow \varepsilon \end{aligned}$$

Running yacc on this grammar produces a `y.output` file that contains the following:

```
0  $accept : S $end
1  S : T
2  |
3  T : '0' '0' T
4  |
...
0: reduce/reduce conflict (reduce 2, reduce 4) on $end
state 0
    $accept : . S $end (0)
    S : . (2)
    T : . (4)

    '0' shift 1
    $end reduce 2

    S goto 2
    T goto 3
```

The following line gives the source of the conflict:

```
0: reduce/reduce conflict (reduce 2, reduce 4) on $end
```

Here, ‘reduce 2’ refers to reducing using grammar rule 2, ‘ $S \rightarrow \varepsilon$ ’; ‘reduce 4’ refers to reducing using grammar rule 4, ‘ $T \rightarrow \varepsilon$ ’; and `$end` refers to EOF. The reason for the conflict, therefore, is that if the next input “token” is EOF (i.e., the input string is ‘ ε ’) then we can either derive this string directly as $S \Rightarrow \varepsilon$ (which would then have us reduce using the rule ‘ $S \rightarrow \varepsilon$ ’ at this point), or we can derive it as $S \Rightarrow T \Rightarrow \varepsilon$ (which would have us first reduce using ‘ $T \rightarrow \varepsilon$ ’).

4 Removing Conflicts

Once we have figured out the actual reason for a conflict, we can try and modify the grammar to remove the conflict without changing the overall language. We can do this either by changing the grammar to remove the source of the conflict, or give yacc additional information to allow it to figure out which of the conflicting actions is the right one to choose, or (if we know what we’re doing) we can leave the conflict alone and rely on yacc’s default action.

It’s important to realize that conflicts cannot always be removed: there are some grammars for which, no matter what we do, we cannot remove all conflicts while preserving equivalence of grammars. The approaches outlined below are therefore heuristics that often work, though they may not always do so.

4.1 Grammar Restructuring

In the case of Grammar 1 above, the problem arises due to the rule ‘ $S \rightarrow \varepsilon$ ’ in the recursion. Given this understanding, we can try to either get rid of this rule, or move it out of the context where the conflict



Figure 1: Parsing choices for expressions

arises. In this particular example, the problem arises because in parser state 1, we can't tell, in the item ' $S \rightarrow \mathbf{0} \bullet S \mathbf{0}$ ', whether an input ' $\mathbf{0}$ ' comes from the S (\Rightarrow shift action) or is the ' $\mathbf{0}$ ' at the end of the item (\Rightarrow reduce action), so we can restructure the grammar as follows to eliminate this source of confusion:

$$\begin{aligned} S &\rightarrow \mathbf{0} \mathbf{0} S \\ S &\rightarrow \varepsilon \end{aligned}$$

With Grammar 2, the problem is that the string ' ε ' can be derived in two ways. One of these, using ' $S \rightarrow \varepsilon$ ', is unnecessary. Getting rid of this rule removes the conflict.

4.2 Providing Additional Information to Yacc

Ambiguities resulting from a lack of information about operator precedences and associativities are a common source of shift/reduce conflicts. These typically arise in situations involving expressions. They can be eliminated using declarations such as `%left` and `%right` (and sometimes `%prec`) to operator precedences and associativities.

As an example, consider the following simple grammar for expressions:

Grammar 3:

$$\begin{aligned} E &\rightarrow E + E \\ E &\rightarrow \text{id} \end{aligned}$$

Running yacc on this gives the following:

```

0  $accept : E $end
1  E : E '+' E
2      | id
...
4: shift/reduce conflict (shift 3, reduce 1) on '+'
state 4
    E : E . '+' E (1)
    E : E '+' E . (1)

    '+' shift 3
    $end reduce 1

```

Here, when the parser sees the token ' $+$ ', it can't decide between the following choices:

1. Assume that the ' $+$ ' comes from the item ' $E \rightarrow E \bullet + E$ ', as indicated by the arrow in Figure 1(a). In this case the parser needs to see the rest of the right hand side of this production before it can

reduce, so the appropriate action is to shift. This choice has the effect of making the operator ‘+’ right-associative.

2. Assume that the ‘+’ comes from a different recursive level of the variable E , as indicated by the arrow shown in Figure 1(b). This means that the current recursive level of E is finished, corresponding to the item ‘ $E \rightarrow E + E \bullet$ ’, and that the parser should carry out a reduce action using rule 1, i.e., ‘ $E \rightarrow E + E$ ’.

It is possible to restructure the grammar to remove this conflict (the details can be found in compiler texts under the heading of *Top-down Parsing*). A simpler alternative is to simply add a declaration to the yacc input file indicating the desired associativity of the ‘+’ operator.